

Tips and Tricks to Create In-text Tables in Clinical Trial Reporting Using SAS

Md. Jabrul Islam, Shafi Consultancy Bangladesh
Shafi Chowdhury, Shafi Consultancy Limited
Rafi Rahi, Shafi Consultancy Limited

ABSTRACT

Although In-text tables were produced by Medical Writers in the past, increasingly the task is being handed over to the programmers. In some cases they are pre-planned and produced at the same time as when the other TLFs are produced for the CTR appendix, however, in most cases they are still defined towards the end of the trial and this can be really time consuming at the busiest period of a programmer's resource.

This paper will show a step-by-step procedure on how to create an In-text table using SAS ODS RTF and directly copy it into an MS Word document with PROC REPORT and PROC TEMPLATE so that it is in the structure required by the Medical Writers. This will arm the programmers with all the tools to make quick updates as and when required and set up templates that can be available for use in all trials.

INTRODUCTION

Programmers used to create tables in Word for Clinical Trial Reports (CTR) by using the SAS ODS RTF destination. Although there are many features of SAS ODS RTF, programmers struggle to utilize its benefits for generating efficient output. In this paper, we will present some basic methods to help the programmers create tables in Word and also demonstrate the process for using the PROC TEMPLATE with ODS RTF and PROC REPORT.

The key points to be discussed in this paper are:

- Syntax and Different statements of PROC TEMPLATE
- Assigning Header, Title and Footnote of Word table
- Page numbering (e.g. Page X of Y)

EXAMPLE 1: SUMMARY OF BASELINE CHARACTERISTICS

PROC TEMPLATE

In the template, we are creating a new style template named RTF1 using the parent style RTF from the SAS catalog. In the new style template, we defined all the attributes needed for the table in Word, e.g. title and footnote fonts, body fonts, attributes for row headings and column headings, background color, link color, internal and external border color and thickness, and page settings for display result.

PROC TEMPLATE;

```
DEFINE STYLE Styles.RTF1;          /* Defining Template Name */
PARENT = Styles.RTF;              /* Parent Style Template */

REPLACE fonts /
  /* Titles from TITLE statements */
  'TitleFont' = ("Calibri, Arial, Times Roman",12pt)
  'StrongFont' = ("Calibri, Times Roman",10pt, Bold)
```

PhUSE 2017

```
'EmphasisFont' = ("Calibri, Times Roman",10pt, Italic)
'headingEmphasisFont' = ("Calibri, Times Roman",11pt, Bold
                        Italic)
/* Table column and row headings */
'headingFont' = ("Arial, Times Roman",10pt, Bold)
/* Data in table cells */
'docFont' = ("Arial, Times Roman",10pt)
/* Footnotes from FOOTNOTE statements */
'footFont' = ("Calibri, Times Roman",10pt)
'FixedEmphasisFont' = ("Courier",9pt, Italic)
'FixedStrongFont' = ("Courier",9pt, Bold)
'FixedHeadingFont' = ("Courier",9pt, Bold)
'BatchFixedFont' = ("Courier",6.7pt)
'FixedFont' = ("Courier",9pt);

REPLACE color_list /
  'link' = blue           /* links */
  'bgH' = white          /* row and column header background */
  'fg' = black           /* text color */
  'bg' = white;         /* page background color */;

CLASS Header /          /* Header Attributes */
  foreground = black
  verticalalign = top;

CLASS TitlesAndFooters / /* Title and Footnote Attributes */
  font = Fonts('TitleFont')
  color = color_list('link')
  textalign = left
  frame = void;
REPLACE Body from Document /
  /* Page setting for display results */
  bottommargin = .7in
  topmargin = .7in
  rightmargin = .7in
  leftmargin = .7in
  protectspecialchars = off;

REPLACE Table from Output /
  frame = box           /* outside borders: void, box, above/below,
                        vsides/hsides, lhs/rhs */
  rules = all          /* internal borders: none,all,cols,rows,groups */
  borderwidth = 1pt
  bordercolor = color_list('fg')
  width = 100%        /* the width of the borders and rules */
  asis = on;

END;
RUN;
```

If anyone wishes to see all the SAS built-in style templates then they can find these using the following code segment:

```
PROC TEMPLATE;
  LIST styles;
RUN;
```

ODS RTF AND PROC REPORT

The following program will create the baseline characteristics summary table named t_demo.rtf (Figure-1) using the above template (styles.rtf1) and the result from the PROC REPORT.

PhUSE 2017

```
* Define page orientation, option to not print date and page number *;
* ODS destination is closed for the Listing *;
* An escape character is specified to perform in-line formatting *;

OPTIONS NODATE NONUMBER ORIENTATION=PORTRAIT;
ODS LISTING CLOSE;
ODS ESCAPECHAR = '^';

* Specify destination path for RTF file and select a style template *;
* Define device type for RTF (only supports: SASSEMFM, PNG and JPEG) *;
* Define titles, footnotes and page numbers (i.e. page x of y)*;

ODS RTF FILE="%pathname./t_demo.rtf" STYLE=styles.rtf1 DEVICE=SASSEMFM;
TITLE 'Table: Patients baseline characteristics';
FOOTNOTE J=CENTER 'Page ^{thispage} of ^{lastpage}';

* Define PROC REPORT procedure statements *;
* Include conditional format for specific rows, columns, footnotes *;

PROC REPORT DATA=final HEADLINE LS=140 PS=50 SPACING=3 SPLIT='$';

    COLUMN (decodeord block decode col1 col2 coltot);

    DEFINE decodeord / ORDER ORDER=INTERNAL NOPRINT;
    DEFINE block      / ORDER ORDER=INTERNAL NOPRINT;
    DEFINE decode     / ' ' FLOW STYLE={WIDTH=27%};
    DEFINE col1       / "DummyRed$(N =&col1.)" WIDTH=26 STYLE={WIDTH=12%};
    DEFINE col2       / "DummyGreen$(N =&col2.)" WIDTH=26 STYLE={WIDTH=12%};
    DEFINE coltot     / "Total$(N =&coltotal.)" WIDTH=26 STYLE={WIDTH=10%};

    COMPUTE decode;
        IF decodeord=0 THEN
            CALL DEFINE (_col_, "STYLE", "STYLE={FONT_WEIGHT=BOLD}");

        IF decode =: " " THEN
            CALL DEFINE (_col_, "Style", "STYLE={INDENT=0.2in}");

        IF decode =: "  " THEN
            CALL DEFINE (_col_, "Style", "STYLE={INDENT=0.4in}");

        IF block ne 1 THEN
            CALL DEFINE (_row_, "STYLE", "STYLE={BORDERBOTTOMCOLOR=white}");
    ENDCOMP;

    COMPUTE AFTER/
        STYLE= {BORDERLEFTCOLOR=white BORDERRIGHTCOLOR=white
                BORDERBOTTOMCOLOR=white};
        LINE @0 "Sources: ADSL";
    ENDCOMP;
RUN;

* Close ODS destination for RTF and open destination for Listing *;
* Reset options for date and page number *;

ODS RTF CLOSE;
ODS LISTING;
OPTIONS DATE NUMBER;
```

PhUSE 2017

OUTPUT OF EXAMPLE TABLE - 1

Using the updated template above, ODS RTF destination and PROC REPORT, the following Word table is produced.

Table: Patients baseline characteristics

	DummyRed (N =136)	DummyGreen (N =166)	Total (N =302)
Age (years), mean (SD)	42.7 (12.67)	44.0 (13.25)	43.4 (12.99)
Weight (kg)			
Mean (SD)	85.8 (20.30)	89.4 (24.75)	87.8 (22.90)
Median (Min, Max)	83.5 (47, 149)	83.0 (48, 200)	83.5 (47, 200)
Weight category			
<= 100 kg	104 (77.0)	121 (72.9)	225 (74.8)
> 100 kg	31 (23.0)	45 (27.1)	76 (25.2)
<= 90 kg	83 (61.5)	103 (62.0)	186 (61.8)
> 90 kg	52 (38.5)	63 (38.0)	115 (38.2)
Gender			
Female	46 (33.8)	54 (32.5)	100 (33.1)
Male	90 (66.2)	112 (67.5)	202 (66.9)
BMI, mean (SD)	28.8 (5.55)	29.7 (6.97)	29.3 (6.38)
Country			
Europe (Total)	111 (81.6)	139 (83.7)	250 (82.8)
Western EU	83 (61.0)	106 (63.9)	189 (62.6)
Eastern EU	28 (20.6)	33 (19.9)	61 (20.2)
Non-EU	25 (18.4)	27 (16.3)	52 (17.2)

Sources: ADSL;

Figure-1

PhUSE 2017

EXAMPLE 2: SUMMARY OF BASELINE CHARACTERISTICS WITH UPDATED TEMPLATE

PROC TEMPLATE

In the PROC TEMPLATE we will keep all of the above statements the same except the following red marks:

```
CLASS Header /                               /* Header Attributes */
  foreground = White
  verticalalign = top;

REPLACE color_list /
  'link' = blue                               /* links */
  'bgH' = cx4F81BD                           /* row and column header background */
  'fg' = black                               /* text color */
  'bg' = white;                              /* page background color */;

REPLACE Table from Output /
  frame = box                                /* outside borders: void, box, above/below,
  vsides/hsides, lhs/rhs */
  rules = all                               /* internal borders: none, all, cols, rows, groups */
  borderwidth = 1pt                         /* the width of the borders and rules */
  bordercolor = color_list('bgH')
  width = 100%
  asis = on;
```

OUTPUT OF EXAMPLE TABLE - 2

Using the updated statements of PROC TEMPLATE above the following Word table will be produced.

Table: Patients baseline characteristics

	DummyRed (N =136)	DummyGreen (N =166)	Total (N =302)
Age (years), mean (SD)	42.7 (12.67)	44.0 (13.25)	43.4 (12.99)
Weight (kg)			
Mean (SD)	85.8 (20.30)	89.4 (24.75)	87.8 (22.90)
Median (Min, Max)	83.5 (47, 149)	83.0 (48, 200)	83.5 (47, 200)
Weight category			
<= 100 kg	104 (77.0)	121 (72.9)	225 (74.8)
> 100 kg	31 (23.0)	45 (27.1)	76 (25.2)
<= 90 kg	83 (61.5)	103 (62.0)	186 (61.8)
> 90 kg	52 (38.5)	63 (38.0)	115 (38.2)
Gender			
Female	46 (33.8)	54 (32.5)	100 (33.1)
Male	90 (66.2)	112 (67.5)	202 (66.9)
BMI, mean (SD)	28.8 (5.55)	29.7 (6.97)	29.3 (6.38)
Country			
Europe (Total)	111 (81.6)	139 (83.7)	250 (82.8)
Western EU	83 (61.0)	106 (63.9)	189 (62.6)
Eastern EU	28 (20.6)	33 (19.9)	61 (20.2)
non-EU	25 (18.4)	27 (16.3)	52 (17.2)

Sources: ADSL;

Figure-2

CONCLUSION

In this paper, we have shown how easily the In-text tables can be created for a CTR. By following the examples provided in this paper, a user can produce the In-text tables within a short period of time by using the TEMPLATE procedure rather than by designing the table in the REPORT procedure. Also, this paper will help users to customize the header, title, footnotes, page numbering (e.g. Page x of y), results attributes, as well as the structure for different layouts of Word tables for clinical trials.

REFERENCES

Carey G. Smoak, "*Creating Word Tables using PROC REPORT and ODS RTF*".
Bob Hull, "*Now There Is an Easy Way to Get to Word, Just Use PROC TEMPLATE, PROC REPORT, and ODS RTF*".
Carol Matthews and Elena Kalchenko, "*Pretty Please?! Making RTF Output "Pretty" with SAS*".

CONTACT DETAILS

If you have any questions or comments please do not hesitate to contact us. Please contact the authors at:

Md. Jabrul Islam
Senior Statistical Programmer
Shafi Consultancy Bangladesh
Email: jabrul@shaficonsultancy.com

Shafi Chowdhury
SAS Consultant & Managing Director
Shafi Consultancy Limited
Email: shafi@shaficonsultancy.com

Rafi Rahi
Senior Statistical Programmer
Shafi Consultancy Limited
Email: rafi@shaficonsultancy.com